# Преобразование Бэклунда и формула суперпозиции для синус-Гордона

Проверка подстановок приводящих уравнения, преобразование Бэклунда и формулу суперпозиции к рациональному виду. Формула суперпозиции записывается в виде $F(q_{m,n},\, q_{m+1,n},\, q_{m,n+1},\, q_{m+1,n+1}) = 0$, где $F$ – многочлен, линейный по каждой переменной, поэтому это уравнение легко разрешается относительно любой из них.

## $u_{xy} = \sinh u$

$$u_{xy} = \sinh u, \quad u_{n+1,x} + u_{n,x} = 2\, a_n \sinh \frac{u_{n+1} - u_n}{2}, \quad u_{n+1,y} - u_{n,y} = \frac{2}{a_n} \sinh \frac{u_{n+1} + u_n}{2}$$

Преобразование Бэклунда:

In[1]:=
```
u[n_] := U_n[x, y]
eqx := D[u[n + 1] + u[n], x] - 2 a[n] Sinh[(u[n + 1] - u[n])/2]
eqy := D[u[n + 1] - u[n], y] - 2/a[n] Sinh[(u[n + 1] + u[n])/2]
Simplify[(D[eqx, y] - D[eqy, x])/2 /. Solve[{eqx, eqy} == 0, {D[u[n + 1], x], D[u[n + 1], y]}][[1]]]
Simplify[(D[eqx, y] + D[eqy, x])/2 /. Solve[{eqx, eqy} == 0, {D[u[n], x], D[u[n], y]}][[1]]]
```

Out[4]= $-\mathrm{Sinh}[U_n[x, y]] + U_n^{(1,1)}[x, y]$

Out[5]= $-\mathrm{Sinh}[U_{1+n}[x, y]] + U_{1+n}^{(1,1)}[x, y]$

Рационализирующая подстановка:

In[6]:=
```
u[n_] := 2 Log[q_n[x, y]]
Simplify[(D[eqx, y] - D[eqy, x])/2 /. Solve[{eqx, eqy} == 0,
    {D[q_{n+1}[x, y], x], D[q_{n+1}[x, y], y]}][[1]]];
Collect[%/Coefficient[%, D[q_n[x, y], x, y]], q_n^(__)[x, y], Factor]


Simplify[(D[eqx, y] + D[eqy, x])/2 /. Solve[{eqx, eqy} == 0,
    {D[q_n[x, y], x], D[q_n[x, y], y]}][[1]]];
Collect[%/Coefficient[%, D[q_{n+1}[x, y], x, y]], q_{n+1}^(__)[x, y], Factor]
```

Out[8]= $$-\frac{(-1 + q_n[x, y])\,(1 + q_n[x, y])\,\left(1 + q_n[x, y]^2\right)}{4\, q_n[x, y]} - \frac{q_n^{(0,1)}[x, y]\, q_n^{(1,0)}[x, y]}{q_n[x, y]} + q_n^{(1,1)}[x, y]$$

Out[10]=

$$-\frac{(-1 + q_{1+n}[x, y])\,(1 + q_{1+n}[x, y])\,\left(1 + q_{1+n}[x, y]^2\right)}{4\, q_{1+n}[x, y]} - \frac{q_{1+n}^{(0,1)}[x, y]\, q_{1+n}^{(1,0)}[x, y]}{q_{1+n}[x, y]} + q_{1+n}^{(1,1)}[x, y]$$

Одевающая цепочка в рациональной форме:

In[11]:=
```
qeqx = Collect[TrigExpand[eqx], q_^(__)[x, y], Factor]
qeqy = Collect[TrigExpand[eqy], q_^(__)[x, y], Factor]
```

Out[11]=

$$\frac{a[n]\,(q_n[x,y]-q_{1+n}[x,y])\,(q_n[x,y]+q_{1+n}[x,y])}{q_n[x,y]\,q_{1+n}[x,y]}+\frac{2\,q_n^{(1,0)}[x,y]}{q_n[x,y]}+\frac{2\,q_{1+n}^{(1,0)}[x,y]}{q_{1+n}[x,y]}$$

Out[12]=

$$-\frac{(-1+q_n[x,y]\,q_{1+n}[x,y])\,(1+q_n[x,y]\,q_{1+n}[x,y])}{a[n]\,q_n[x,y]\,q_{1+n}[x,y]}-\frac{2\,q_n^{(0,1)}[x,y]}{q_n[x,y]}+\frac{2\,q_{1+n}^{(0,1)}[x,y]}{q_{1+n}[x,y]}$$

Вводим второй индекс:

In[13]:=
```
neqx = qeqx /. q_n_ :→ q_m,n
meqx = qeqx /. n → m /. q_m_ :→ q_m,n

neqy = qeqy /. q_n_ :→ q_m,n
meqy = qeqy /. n → m /. q_m_ :→ q_m,n
```

Out[13]=

$$\frac{a[n]\,(q_{m,n}[x,y]-q_{m,1+n}[x,y])\,(q_{m,n}[x,y]+q_{m,1+n}[x,y])}{q_{m,n}[x,y]\,q_{m,1+n}[x,y]}+\frac{2\,q_{m,n}^{(1,0)}[x,y]}{q_{m,n}[x,y]}+\frac{2\,q_{m,1+n}^{(1,0)}[x,y]}{q_{m,1+n}[x,y]}$$

Out[14]=

$$\frac{a[m]\,(q_{m,n}[x,y]-q_{1+m,n}[x,y])\,(q_{m,n}[x,y]+q_{1+m,n}[x,y])}{q_{m,n}[x,y]\,q_{1+m,n}[x,y]}+\frac{2\,q_{m,n}^{(1,0)}[x,y]}{q_{m,n}[x,y]}+\frac{2\,q_{1+m,n}^{(1,0)}[x,y]}{q_{1+m,n}[x,y]}$$

Out[15]=

$$-\frac{(-1+q_{m,n}[x,y]\,q_{m,1+n}[x,y])\,(1+q_{m,n}[x,y]\,q_{m,1+n}[x,y])}{a[n]\,q_{m,n}[x,y]\,q_{m,1+n}[x,y]}-\frac{2\,q_{m,n}^{(0,1)}[x,y]}{q_{m,n}[x,y]}+\frac{2\,q_{m,1+n}^{(0,1)}[x,y]}{q_{m,1+n}[x,y]}$$

Out[16]=

$$-\frac{(-1+q_{m,n}[x,y]\,q_{1+m,n}[x,y])\,(1+q_{m,n}[x,y]\,q_{1+m,n}[x,y])}{a[m]\,q_{m,n}[x,y]\,q_{1+m,n}[x,y]}-\frac{2\,q_{m,n}^{(0,1)}[x,y]}{q_{m,n}[x,y]}+\frac{2\,q_{1+m,n}^{(0,1)}[x,y]}{q_{1+m,n}[x,y]}$$

Комбинируем уравнения, чтобы уничтожить производные, это даёт формулу суперпозиции. Для краткости убираем $x$, $y$. Формула суперпозиции факторизуется. Знаменатель отбрасываем. В числителе один множитель не зависит от параметров и можно проверить, что если приравнять его 0, то это уравнение не будет совместно с дифференцированиями по $x$ и по $y$. Поэтому этот множитель можно отбросить. А второй множитель даёт уравнение, совместное с дифференцированиями – это и есть формула суперпозиции. В результате получается уравнение, линейное по каждой из переменной $q_{m,n}$, $q_{m+1,n}$, $q_{m,n+1}$, $q_{m+1,n+1}$.

In[17]:=

```
(neqx /. m → m + 1) - (meqx /. n → n + 1) - meqx + neqx

Factor[% /. q_m__[x, y] :→ q_m]

F1 = Numerator[%] [[1]]
F2 = Numerator[%%] [[2]]
```

Out[17]=

$$\frac{a[n]\,(q_{m,n}[x,y]-q_{m,1+n}[x,y])\,(q_{m,n}[x,y]+q_{m,1+n}[x,y])}{q_{m,n}[x,y]\,q_{m,1+n}[x,y]}-\frac{a[m]\,(q_{m,n}[x,y]-q_{1+m,n}[x,y])\,(q_{m,n}[x,y]+q_{1+m,n}[x,y])}{q_{m,n}[x,y]\,q_{1+m,n}[x,y]}-\frac{a[m]\,(q_{m,1+n}[x,y]-q_{1+m,1+n}[x,y])\,(q_{m,1+n}[x,y]+q_{1+m,1+n}[x,y])}{q_{m,1+n}[x,y]\,q_{1+m,1+n}[x,y]}+\frac{a[n]\,(q_{1+m,n}[x,y]-q_{1+m,1+n}[x,y])\,(q_{1+m,n}[x,y]+q_{1+m,1+n}[x,y])}{q_{1+m,n}[x,y]\,q_{1+m,1+n}[x,y]}$$

Out[18]=

$$\frac{(q_{m,1+n}\,q_{1+m,n}+q_{m,n}\,q_{1+m,1+n})\,(-a[m]\,q_{m,n}\,q_{m,1+n}+a[n]\,q_{m,n}\,q_{1+m,n}-a[n]\,q_{m,1+n}\,q_{1+m,1+n}+a[m]\,q_{1+m,n}\,q_{1+m,1+n})}{q_{m,n}\,q_{m,1+n}\,q_{1+m,n}\,q_{1+m,1+n}}$$

Out[19]=

$$q_{m,1+n}\,q_{1+m,n}+q_{m,n}\,q_{1+m,1+n}$$

Out[20]=

$$-a[m]\,q_{m,n}\,q_{m,1+n}+a[n]\,q_{m,n}\,q_{1+m,n}-a[n]\,q_{m,1+n}\,q_{1+m,1+n}+a[m]\,q_{1+m,n}\,q_{1+m,1+n}$$

Проверим совместность, дифференцируя в силу цепочек. Для этого сначала выразим производные:

In[21]:=
```
{neqx, neqx /. m → m + 1, meqx} == 0;
solx = Solve[%, {D[q_{m,n+1}[x, y], x], D[q_{m+1,n}[x, y], x], D[q_{m+1,n+1}[x, y], x]}][[1]]

{neqy, neqy /. m → m + 1, meqy} == 0;
soly = Solve[%, {D[q_{m,n+1}[x, y], y], D[q_{m+1,n}[x, y], y], D[q_{m+1,n+1}[x, y], y]}][[1]]
```

Out[22]=

$$\Big\{q_{m,1+n}{}^{(1,0)}[x,y] \to -\frac{a[n]\, q_{m,n}[x,y]^2 - a[n]\, q_{m,1+n}[x,y]^2 + 2\, q_{m,1+n}[x,y]\, q_{m,n}{}^{(1,0)}[x,y]}{2\, q_{m,n}[x,y]},$$

$$q_{1+m,n}{}^{(1,0)}[x,y] \to -\frac{a[m]\, q_{m,n}[x,y]^2 - a[m]\, q_{1+m,n}[x,y]^2 + 2\, q_{1+m,n}[x,y]\, q_{m,n}{}^{(1,0)}[x,y]}{2\, q_{m,n}[x,y]},$$

$$q_{1+m,1+n}{}^{(1,0)}[x,y] \to -\frac{1}{2\, q_{m,n}[x,y]\, q_{1+m,n}[x,y]} \Big(a[n]\, q_{m,n}[x,y]\, q_{1+m,n}[x,y]^2 - a[m]\, q_{m,n}[x,y]^2\, q_{1+m,1+n}[x,y] + a[m]\, q_{1+m,n}[x,y]^2\, q_{1+m,1+n}[x,y] - a[n]\, q_{m,n}[x,y]\, q_{1+m,1+n}[x,y]^2 - 2\, q_{1+m,n}[x,y]\, q_{1+m,1+n}[x,y]\, q_{m,n}{}^{(1,0)}[x,y]\Big)\Big\}$$

Out[24]=

$$\Big\{q_{m,1+n}{}^{(0,1)}[x,y] \to -\frac{1 - q_{m,n}[x,y]^2\, q_{m,1+n}[x,y]^2 - 2\, a[n]\, q_{m,1+n}[x,y]\, q_{m,n}{}^{(0,1)}[x,y]}{2\, a[n]\, q_{m,n}[x,y]},$$

$$q_{1+m,n}{}^{(0,1)}[x,y] \to -\frac{1 - q_{m,n}[x,y]^2\, q_{1+m,n}[x,y]^2 - 2\, a[m]\, q_{1+m,n}[x,y]\, q_{m,n}{}^{(0,1)}[x,y]}{2\, a[m]\, q_{m,n}[x,y]},$$

$$q_{1+m,1+n}{}^{(0,1)}[x,y] \to -\Big(\big(a[m]\, q_{m,n}[x,y] + a[n]\, q_{1+m,1+n}[x,y] - a[n]\, q_{m,n}[x,y]^2\, q_{1+m,n}[x,y]^2\, q_{1+m,1+n}[x,y] - a[m]\, q_{m,n}[x,y]\, q_{1+m,n}[x,y]^2\, q_{1+m,1+n}[x,y]^2 - 2\, a[m] \times a[n]\, q_{1+m,n}[x,y]\, q_{1+m,1+n}[x,y]\, q_{m,n}{}^{(0,1)}[x,y]\big) \Big/ \big(2\, a[m] \times a[n]\, q_{m,n}[x,y]\, q_{1+m,n}[x,y]\big)\Big)\Big\}$$

Теперь продифференцируем первый множитель. Нуля не получается

In[25]:=
```
f1 = F1 /. q_{m__} :> q_m[x, y]
Factor[D[f1, x] /. solx /. Solve[f1 == 0, q_{m+1,n+1}[x, y]][[1]]]

Factor[D[f1, y] /. soly /. Solve[f1 == 0, q_{m+1,n+1}[x, y]][[1]]]
```

Out[25]=

$$q_{m,1+n}[x,y]\, q_{1+m,n}[x,y] + q_{m,n}[x,y]\, q_{1+m,1+n}[x,y]$$

Out[26]=

$$-\frac{1}{q_{m,n}[x,y]} \Big(a[m]\, q_{m,n}[x,y]^2\, q_{m,1+n}[x,y] + a[n]\, q_{m,n}[x,y]^2\, q_{1+m,n}[x,y] - a[n]\, q_{m,1+n}[x,y]^2\, q_{1+m,n}[x,y] - a[m]\, q_{m,1+n}[x,y]\, q_{1+m,n}[x,y]^2 + 4\, q_{m,1+n}[x,y]\, q_{1+m,n}[x,y]\, q_{m,n}{}^{(1,0)}[x,y]\Big)$$

Out[27]=

$$\frac{\big(-1 + q_{m,1+n}[x,y]\, q_{1+m,n}[x,y]\big)\big(1 + q_{m,1+n}[x,y]\, q_{1+m,n}[x,y]\big)\big(q_{m,n}[x,y]^2 + q_{1+m,n}[x,y]^2\big)}{2\, a[n]\, q_{m,n}[x,y]\, q_{1+m,n}[x,y]}$$

А со вторым множителем – все в порядке. Он дает уравнение совместное с цепочками.

In[28]:=
```
f2 = F2 /. q_m__ :> q_m[x, y]
Factor[D[f2, x] /. solx /. Solve[f2 == 0, q_{m+1,n+1}[x, y]][[1]]]
Factor[D[f2, y] /. soly /. Solve[f2 == 0, q_{m+1,n+1}[x, y]][[1]]]
```

Out[28]=

$$-a[m]\, q_{m,n}[x,y]\, q_{m,1+n}[x,y] + a[n]\, q_{m,n}[x,y]\, q_{1+m,n}[x,y] - a[n]\, q_{m,1+n}[x,y]\, q_{1+m,1+n}[x,y] + a[m]\, q_{1+m,n}[x,y]\, q_{1+m,1+n}[x,y]$$

Out[29]=

0

Out[30]=

0

## $u_{xy} = \sin u$

$u_{xy} = \sin u, \quad u_{n+1,x} + u_{n,x} = 2\, a_n \sin \frac{u_{n+1} - u_n}{2}, \quad u_{n+1,y} - u_{n,y} = \frac{2}{a_n} \sin \frac{u_{n+1} + u_n}{2}$

Преобразование Бэклунда:

In[31]:=
```
u[n_] := U_n[x, y]
eqx := D[u[n + 1] + u[n], x] - 2 a[n] Sin[(u[n + 1] - u[n])/2]
eqy := D[u[n + 1] - u[n], y] - 2/a[n] Sin[(u[n + 1] + u[n])/2]
Simplify[(D[eqx, y] - D[eqy, x])/2 /. Solve[{eqx, eqy} == 0, {D[u[n + 1], x], D[u[n + 1], y]}][[1]]]
Simplify[(D[eqx, y] + D[eqy, x])/2 /. Solve[{eqx, eqy} == 0, {D[u[n], x], D[u[n], y]}][[1]]]
```

Out[34]=

$-\mathrm{Sin}[U_n[x,y]] + U_n^{(1,1)}[x,y]$

Out[35]=

$-\mathrm{Sin}[U_{1+n}[x,y]] + U_{1+n}^{(1,1)}[x,y]$

Рационализирующая подстановка:

In[36]:=
```
u[n_] := 4 ArcTan[q_n[x, y]]
Simplify[(D[eqx, y] - D[eqy, x])/2 /. Solve[{eqx, eqy} == 0,
      {D[q_{n+1}[x, y], x], D[q_{n+1}[x, y], y]}][[1]]];
Collect[%/Coefficient[%, D[q_n[x, y], x, y]], q_n^(__)[x, y], Factor]


Simplify[(D[eqx, y] + D[eqy, x])/2 /. Solve[{eqx, eqy} == 0,
      {D[q_n[x, y], x], D[q_n[x, y], y]}][[1]]];
Collect[%/Coefficient[%, D[q_{n+1}[x, y], x, y]], q_{n+1}^(__)[x, y], Factor]
```

Out[38]=

$$\frac{(-1+q_n[x,y])\,q_n[x,y]\,(1+q_n[x,y])}{1+q_n[x,y]^2}-\frac{2\,q_n[x,y]\,q_n^{(0,1)}[x,y]\,q_n^{(1,0)}[x,y]}{1+q_n[x,y]^2}+q_n^{(1,1)}[x,y]$$

Out[40]=

$$\frac{(-1+q_{1+n}[x,y])\,q_{1+n}[x,y]\,(1+q_{1+n}[x,y])}{1+q_{1+n}[x,y]^2}-\frac{2\,q_{1+n}[x,y]\,q_{1+n}^{(0,1)}[x,y]\,q_{1+n}^{(1,0)}[x,y]}{1+q_{1+n}[x,y]^2}+q_{1+n}^{(1,1)}[x,y]$$

Одевающая цепочка в рациональной форме:

In[41]:=
```
qeqx = Collect[TrigExpand[eqx], q_^(__)[x, y], Factor]
qeqy = Collect[TrigExpand[eqy], q_^(__)[x, y], Factor]
```

Out[41]=

$$\frac{4\,a[n]\,(q_n[x,y]-q_{1+n}[x,y])\,(1+q_n[x,y]\,q_{1+n}[x,y])}{\left(1+q_n[x,y]^2\right)\left(1+q_{1+n}[x,y]^2\right)}+\frac{4\,q_n^{(1,0)}[x,y]}{1+q_n[x,y]^2}+\frac{4\,q_{1+n}^{(1,0)}[x,y]}{1+q_{1+n}[x,y]^2}$$

Out[42]=

$$\frac{4\,(q_n[x,y]+q_{1+n}[x,y])\,(-1+q_n[x,y]\,q_{1+n}[x,y])}{a[n]\left(1+q_n[x,y]^2\right)\left(1+q_{1+n}[x,y]^2\right)}-\frac{4\,q_n^{(0,1)}[x,y]}{1+q_n[x,y]^2}+\frac{4\,q_{1+n}^{(0,1)}[x,y]}{1+q_{1+n}[x,y]^2}$$

Вводим второй индекс:

In[43]:=
```
neqx = qeqx /. q_n_ :> q_{m,n}
meqx = qeqx /. n -> m /. q_m_ :> q_{m,n}

neqy = qeqy /. q_n_ :> q_{m,n}
meqy = qeqy /. n -> m /. q_m_ :> q_{m,n}
```

Out[43]=

$$\frac{4\,a[n]\,(q_{m,n}[x,y]-q_{m,1+n}[x,y])\,(1+q_{m,n}[x,y]\,q_{m,1+n}[x,y])}{\left(1+q_{m,n}[x,y]^2\right)\left(1+q_{m,1+n}[x,y]^2\right)}+\frac{4\,q_{m,n}^{(1,0)}[x,y]}{1+q_{m,n}[x,y]^2}+\frac{4\,q_{m,1+n}^{(1,0)}[x,y]}{1+q_{m,1+n}[x,y]^2}$$

Out[44]=

$$\frac{4\,a[m]\,(q_{m,n}[x,y]-q_{1+m,n}[x,y])\,(1+q_{m,n}[x,y]\,q_{1+m,n}[x,y])}{\left(1+q_{m,n}[x,y]^2\right)\left(1+q_{1+m,n}[x,y]^2\right)}+\frac{4\,q_{m,n}^{(1,0)}[x,y]}{1+q_{m,n}[x,y]^2}+\frac{4\,q_{1+m,n}^{(1,0)}[x,y]}{1+q_{1+m,n}[x,y]^2}$$

Out[45]=

$$\frac{4\,(q_{m,n}[x,y]+q_{m,1+n}[x,y])\,(-1+q_{m,n}[x,y]\,q_{m,1+n}[x,y])}{a[n]\left(1+q_{m,n}[x,y]^2\right)\left(1+q_{m,1+n}[x,y]^2\right)}-\frac{4\,q_{m,n}^{(0,1)}[x,y]}{1+q_{m,n}[x,y]^2}+\frac{4\,q_{m,1+n}^{(0,1)}[x,y]}{1+q_{m,1+n}[x,y]^2}$$

Out[46]=

$$\frac{4\,(q_{m,n}[x,y]+q_{1+m,n}[x,y])\,(-1+q_{m,n}[x,y]\,q_{1+m,n}[x,y])}{a[m]\left(1+q_{m,n}[x,y]^2\right)\left(1+q_{1+m,n}[x,y]^2\right)}-\frac{4\,q_{m,n}^{(0,1)}[x,y]}{1+q_{m,n}[x,y]^2}+\frac{4\,q_{1+m,n}^{(0,1)}[x,y]}{1+q_{1+m,n}[x,y]^2}$$

Комбинируем уравнения, чтобы уничтожить производные, это даёт формулу суперпозиции. Для краткости убираем $x$, $y$. Формула суперпозиции факторизуется. Знаменатель отбрасываем. В числителе один множитель не зависит от параметров и можно проверить, что если приравнять его 0, то это уравнение не будет совместно с дифференцированиями по $x$ и по $y$. Поэтому этот множитель можно отбросить. А второй множитель даёт уравнение, совместное с дифференцированиями – это и есть формула суперпозиции.  В результате  получается уравнение, линейное по каждой из переменной $q_{m,n}$, $q_{m+1,n}$, $q_{m,n+1}$, $q_{m+1,n+1}$.

In[47]:=

```
(neqx /. m → m + 1) - (meqx /. n → n + 1) - meqx + neqx

Factor[% / 4 /. q_m__[x, y] :→ q_m]

F1 = Numerator[%] [[2]]
F2 = Numerator[%%] [[1]]
```

Out[47]=

$$\frac{4\,a[n]\,(q_{m,n}[x,y]-q_{m,1+n}[x,y])\,(1+q_{m,n}[x,y]\,q_{m,1+n}[x,y])}{\left(1+q_{m,n}[x,y]^2\right)\left(1+q_{m,1+n}[x,y]^2\right)}-\frac{4\,a[m]\,(q_{m,n}[x,y]-q_{1+m,n}[x,y])\,(1+q_{m,n}[x,y]\,q_{1+m,n}[x,y])}{\left(1+q_{m,n}[x,y]^2\right)\left(1+q_{1+m,n}[x,y]^2\right)}-\frac{4\,a[m]\,(q_{m,1+n}[x,y]-q_{1+m,1+n}[x,y])\,(1+q_{m,1+n}[x,y]\,q_{1+m,1+n}[x,y])}{\left(1+q_{m,1+n}[x,y]^2\right)\left(1+q_{1+m,1+n}[x,y]^2\right)}+\frac{4\,a[n]\,(q_{1+m,n}[x,y]-q_{1+m,1+n}[x,y])\,(1+q_{1+m,n}[x,y]\,q_{1+m,1+n}[x,y])}{\left(1+q_{1+m,n}[x,y]^2\right)\left(1+q_{1+m,1+n}[x,y]^2\right)}$$

Out[48]=

$$((-a[m]\,q_{m,n}+a[n]\,q_{m,n}-a[m]\,q_{m,1+n}-a[n]\,q_{m,1+n}+a[m]\,q_{1+m,n}+a[n]\,q_{1+m,n}-a[m]\,q_{m,n}\,q_{m,1+n}\,q_{1+m,n}+a[n]\,q_{m,n}\,q_{m,1+n}\,q_{1+m,n}+a[m]\,q_{1+m,1+n}-a[n]\,q_{1+m,1+n}-a[m]\,q_{m,n}\,q_{m,1+n}\,q_{1+m,1+n}-a[n]\,q_{m,n}\,q_{m,1+n}\,q_{1+m,1+n}+a[m]\,q_{m,n}\,q_{1+m,n}\,q_{1+m,1+n}+a[n]\,q_{m,n}\,q_{1+m,n}\,q_{1+m,1+n}+a[m]\,q_{m,1+n}\,q_{1+m,n}\,q_{1+m,1+n}-a[n]\,q_{m,1+n}\,q_{1+m,n}\,q_{1+m,1+n})\,(1+q_{m,n}\,q_{m,1+n}+q_{m,n}\,q_{1+m,n}-q_{m,1+n}\,q_{1+m,n}-q_{m,n}\,q_{1+m,1+n}+q_{m,1+n}\,q_{1+m,1+n}+q_{1+m,n}\,q_{1+m,1+n}+q_{m,n}\,q_{m,1+n}\,q_{1+m,n}\,q_{1+m,1+n}))\Big/\left(\left(1+q_{m,n}^2\right)\left(1+q_{m,1+n}^2\right)\left(1+q_{1+m,n}^2\right)\left(1+q_{1+m,1+n}^2\right)\right)$$

Out[49]=

$$1+q_{m,n}\,q_{m,1+n}+q_{m,n}\,q_{1+m,n}-q_{m,1+n}\,q_{1+m,n}-q_{m,n}\,q_{1+m,1+n}+q_{m,1+n}\,q_{1+m,1+n}+q_{1+m,n}\,q_{1+m,1+n}+q_{m,n}\,q_{m,1+n}\,q_{1+m,n}\,q_{1+m,1+n}$$

Out[50]=

$$-a[m]\,q_{m,n}+a[n]\,q_{m,n}-a[m]\,q_{m,1+n}-a[n]\,q_{m,1+n}+a[m]\,q_{1+m,n}+a[n]\,q_{1+m,n}-a[m]\,q_{m,n}\,q_{m,1+n}\,q_{1+m,n}+a[n]\,q_{m,n}\,q_{m,1+n}\,q_{1+m,n}+a[m]\,q_{1+m,1+n}-a[n]\,q_{1+m,1+n}-a[m]\,q_{m,n}\,q_{m,1+n}\,q_{1+m,1+n}-a[n]\,q_{m,n}\,q_{m,1+n}\,q_{1+m,1+n}+a[m]\,q_{m,n}\,q_{1+m,n}\,q_{1+m,1+n}+a[n]\,q_{m,n}\,q_{1+m,n}\,q_{1+m,1+n}+a[m]\,q_{m,1+n}\,q_{1+m,n}\,q_{1+m,1+n}-a[n]\,q_{m,1+n}\,q_{1+m,n}\,q_{1+m,1+n}$$

Проверим совместность, дифференцируя в силу цепочек. Для этого сначала выразим производные:

In[51]:=

```
{neqx, neqx /. m → m + 1, meqx} == 0;
solx = Solve[%, {D[q_{m,n+1}[x, y], x], D[q_{m+1,n}[x, y], x], D[q_{m+1,n+1}[x, y], x]}][[1]]

{neqy, neqy /. m → m + 1, meqy} == 0;
soly = Solve[%, {D[q_{m,n+1}[x, y], y], D[q_{m+1,n}[x, y], y], D[q_{m+1,n+1}[x, y], y]}][[1]]
```

Out[52]=

$$\Big\{q_{m,1+n}{}^{(1,0)}[x,y] \to -\frac{1}{1+q_{m,n}[x,y]^2}\Big(a[n]\,q_{m,n}[x,y] - a[n]\,q_{m,1+n}[x,y] + a[n]\,q_{m,n}[x,y]^2\,q_{m,1+n}[x,y] - a[n]\,q_{m,n}[x,y]\,q_{m,1+n}[x,y]^2 + q_{m,n}{}^{(1,0)}[x,y] + q_{m,1+n}[x,y]^2\,q_{m,n}{}^{(1,0)}[x,y]\Big),$$
$$q_{1+m,n}{}^{(1,0)}[x,y] \to -\frac{1}{1+q_{m,n}[x,y]^2}\Big(a[m]\,q_{m,n}[x,y] - a[m]\,q_{1+m,n}[x,y] + a[m]\,q_{m,n}[x,y]^2\,q_{1+m,n}[x,y] - a[m]\,q_{m,n}[x,y]\,q_{1+m,n}[x,y]^2 + q_{m,n}{}^{(1,0)}[x,y] + q_{1+m,n}[x,y]^2\,q_{m,n}{}^{(1,0)}[x,y]\Big),\ q_{1+m,1+n}{}^{(1,0)}[x,y] \to$$
$$-\Big(\Big(-a[m]\,q_{m,n}[x,y] + a[m]\,q_{1+m,n}[x,y] + a[n]\,q_{1+m,n}[x,y] - a[m]\,q_{m,n}[x,y]^2\,q_{1+m,n}[x,y] + a[n]\,q_{m,n}[x,y]^2\,q_{1+m,n}[x,y] + a[m]\,q_{m,n}[x,y]\,q_{1+m,n}[x,y]^2 - a[n]\,q_{1+m,1+n}[x,y] - a[n]\,q_{m,n}[x,y]^2\,q_{1+m,1+n}[x,y] + a[n]\,q_{1+m,n}[x,y]^2\,q_{1+m,1+n}[x,y] + a[n]\,q_{m,n}[x,y]^2\,q_{1+m,n}[x,y]^2\,q_{1+m,1+n}[x,y] - a[m]\,q_{m,n}[x,y]\,q_{1+m,1+n}[x,y]^2 + a[m]\,q_{1+m,n}[x,y]\,q_{1+m,1+n}[x,y]^2 - a[n]\,q_{1+m,n}[x,y]\,q_{1+m,1+n}[x,y]^2 - a[m]\,q_{m,n}[x,y]^2\,q_{1+m,n}[x,y]\,q_{1+m,1+n}[x,y]^2 - a[n]\,q_{m,n}[x,y]^2\,q_{1+m,n}[x,y]\,q_{1+m,1+n}[x,y]^2 + a[m]\,q_{m,n}[x,y]\,q_{1+m,n}[x,y]^2\,q_{1+m,1+n}[x,y]^2 - q_{m,n}{}^{(1,0)}[x,y] - q_{1+m,n}[x,y]^2\,q_{m,n}{}^{(1,0)}[x,y] - q_{1+m,1+n}[x,y]^2\,q_{m,n}{}^{(1,0)}[x,y] - q_{1+m,n}[x,y]^2\,q_{1+m,1+n}[x,y]^2\,q_{m,n}{}^{(1,0)}[x,y]\Big)\Big/\Big(\big(1+q_{m,n}[x,y]^2\big)\big(1+q_{1+m,n}[x,y]^2\big)\Big)\Big)\Big\}$$

Out[54]=

$$\Big\{q_{m,1+n}{}^{(0,1)}[x,y] \to -\frac{1}{a[n]\big(1+q_{m,n}[x,y]^2\big)}\Big(-q_{m,n}[x,y] - q_{m,1+n}[x,y] + q_{m,n}[x,y]^2\,q_{m,1+n}[x,y] + q_{m,n}[x,y]\,q_{m,1+n}[x,y]^2 - a[n]\,q_{m,n}{}^{(0,1)}[x,y] - a[n]\,q_{m,1+n}[x,y]^2\,q_{m,n}{}^{(0,1)}[x,y]\Big),$$
$$q_{1+m,n}{}^{(0,1)}[x,y] \to -\frac{1}{a[m]\big(1+q_{m,n}[x,y]^2\big)}\Big(-q_{m,n}[x,y] - q_{1+m,n}[x,y] + q_{m,n}[x,y]^2\,q_{1+m,n}[x,y] + q_{m,n}[x,y]\,q_{1+m,n}[x,y]^2 - a[m]\,q_{m,n}{}^{(0,1)}[x,y] - a[m]\,q_{1+m,n}[x,y]^2\,q_{m,n}{}^{(0,1)}[x,y]\Big),$$
$$q_{1+m,1+n}{}^{(0,1)}[x,y] \to -\Big(\Big(-a[n]\,q_{m,n}[x,y] - a[m]\,q_{1+m,n}[x,y] - a[n]\,q_{1+m,n}[x,y] - a[m]\,q_{m,n}[x,y]^2\,q_{1+m,n}[x,y] + a[n]\,q_{m,n}[x,y]^2\,q_{1+m,n}[x,y] + a[n]\,q_{m,n}[x,y]\,q_{1+m,n}[x,y]^2 - a[m]\,q_{1+m,1+n}[x,y] - a[m]\,q_{m,n}[x,y]^2\,q_{1+m,1+n}[x,y] + a[m]\,q_{1+m,n}[x,y]^2\,q_{1+m,1+n}[x,y] + a[m]\,q_{m,n}[x,y]^2\,q_{1+m,n}[x,y]^2\,q_{1+m,1+n}[x,y] - a[n]\,q_{m,n}[x,y]\,q_{1+m,1+n}[x,y]^2 + a[m]\,q_{1+m,n}[x,y]\,q_{1+m,1+n}[x,y]^2 - a[n]\,q_{1+m,n}[x,y]\,q_{1+m,1+n}[x,y]^2 + a[m]\,q_{m,n}[x,y]^2\,q_{1+m,n}[x,y]\,q_{1+m,1+n}[x,y]^2 + a[n]\,q_{m,n}[x,y]^2\,q_{1+m,n}[x,y]\,q_{1+m,1+n}[x,y]^2 + a[n]\,q_{m,n}[x,y]\,q_{1+m,n}[x,y]^2\,q_{1+m,1+n}[x,y]^2 - a[m]\times a[n]\,q_{m,n}{}^{(0,1)}[x,y] - a[m]\times a[n]\,q_{1+m,n}[x,y]^2\,q_{m,n}{}^{(0,1)}[x,y] - a[m]\times a[n]\,q_{1+m,1+n}[x,y]^2\,q_{m,n}{}^{(0,1)}[x,y] - a[m]\times a[n]\,q_{1+m,n}[x,y]^2\,q_{1+m,1+n}[x,y]^2\,q_{m,n}{}^{(0,1)}[x,y]\Big)\Big/\Big(a[m]\times a[n]\big(1+q_{m,n}[x,y]^2\big)\big(1+q_{1+m,n}[x,y]^2\big)\Big)\Big)\Big\}$$

Теперь продифференцируем первый множитель. Нуля не получается

In[55]:=
```
f1 = F1 /. q_m__ ⧴ q_m[x, y]
Factor[D[f1, x] /. solx /. Solve[f1 == 0, q_{m+1,n+1}[x, y]][[1]]]

Factor[D[f1, y] /. soly /. Solve[f1 == 0, q_{m+1,n+1}[x, y]][[1]]]
```

Out[55]=

$$1 + q_{m,n}[x,y]\, q_{m,1+n}[x,y] + q_{m,n}[x,y]\, q_{1+m,n}[x,y] - q_{m,1+n}[x,y]\, q_{1+m,n}[x,y] - q_{m,n}[x,y]\, q_{1+m,1+n}[x,y] + q_{m,1+n}[x,y]\, q_{1+m,1+n}[x,y] + q_{1+m,n}[x,y]\, q_{1+m,1+n}[x,y] + q_{m,n}[x,y]\, q_{m,1+n}[x,y]\, q_{1+m,n}[x,y]\, q_{1+m,1+n}[x,y]$$

Out[56]=

$$\Big(2 \Big(a[m]\, q_{m,n}[x,y] + a[n]\, q_{m,n}[x,y] - a[n]\, q_{m,1+n}[x,y] + a[n]\, q_{m,n}[x,y]^2 q_{m,1+n}[x,y] + a[m]\, q_{m,n}[x,y]\, q_{m,1+n}[x,y]^2 - a[n]\, q_{m,n}[x,y]\, q_{m,1+n}[x,y]^2 - a[m]\, q_{1+m,n}[x,y] + a[m]\, q_{m,n}[x,y]^2 q_{1+m,n}[x,y] - a[m]\, q_{m,1+n}[x,y]^2 q_{1+m,n}[x,y] + a[m]\, q_{m,n}[x,y]^2 q_{m,1+n}[x,y]^2 q_{1+m,n}[x,y] - a[m]\, q_{m,n}[x,y]\, q_{1+m,n}[x,y]^2 + a[n]\, q_{m,n}[x,y]\, q_{1+m,n}[x,y]^2 - a[n]\, q_{m,1+n}[x,y]\, q_{1+m,n}[x,y]^2 + a[n]\, q_{m,n}[x,y]^2 q_{m,1+n}[x,y]\, q_{1+m,n}[x,y]^2 - a[m]\, q_{m,n}[x,y]\, q_{m,1+n}[x,y]^2 q_{1+m,n}[x,y]^2 - a[n]\, q_{m,n}[x,y]\, q_{m,1+n}[x,y]^2 q_{1+m,n}[x,y]^2 + 2\, q_{m,n}^{(1,0)}[x,y] + 2\, q_{m,1+n}[x,y]^2 q_{m,n}^{(1,0)}[x,y] + 2\, q_{1+m,n}[x,y]^2 q_{m,n}^{(1,0)}[x,y] + 2\, q_{m,1+n}[x,y]^2 q_{1+m,n}[x,y]^2 q_{m,n}^{(1,0)}[x,y]\Big)\Big) \Big/ \left(-q_{m,n}[x,y] + q_{m,1+n}[x,y] + q_{1+m,n}[x,y] + q_{m,n}[x,y]\, q_{m,1+n}[x,y]\, q_{1+m,n}[x,y]\right)$$

Out[57]=

$$\left(2 \left(q_{m,1+n}[x,y] + q_{1+m,n}[x,y]\right) \left(1 + q_{m,n}[x,y] - q_{1+m,n}[x,y] + q_{m,n}[x,y]\, q_{1+m,n}[x,y]\right) \left(1 - q_{m,n}[x,y] + q_{1+m,n}[x,y] + q_{m,n}[x,y]\, q_{1+m,n}[x,y]\right) \left(-1 + q_{m,1+n}[x,y]\, q_{1+m,n}[x,y]\right)\right) \Big/ \Big(a[n] \left(-q_{m,n}[x,y] + q_{m,1+n}[x,y] + q_{1+m,n}[x,y] + q_{m,n}[x,y]\, q_{m,1+n}[x,y]\, q_{1+m,n}[x,y]\right) \left(1 + q_{1+m,n}[x,y]^2\right)\Big)$$

А со вторым множителем – все в порядке. Он дает уравнение совместное с цепочками.

In[58]:=
```
f2 = F2 /. q_m__ ⧴ q_m[x, y]
Factor[D[f2, x] /. solx /. Solve[f2 == 0, q_{m+1,n+1}[x, y]][[1]]]
Factor[D[f2, y] /. soly /. Solve[f2 == 0, q_{m+1,n+1}[x, y]][[1]]]
```

Out[58]=

$$-a[m]\, q_{m,n}[x,y] + a[n]\, q_{m,n}[x,y] - a[m]\, q_{m,1+n}[x,y] - a[n]\, q_{m,1+n}[x,y] + a[m]\, q_{1+m,n}[x,y] + a[n]\, q_{1+m,n}[x,y] - a[m]\, q_{m,n}[x,y]\, q_{m,1+n}[x,y]\, q_{1+m,n}[x,y] + a[n]\, q_{m,n}[x,y]\, q_{m,1+n}[x,y]\, q_{1+m,n}[x,y] + a[m]\, q_{1+m,1+n}[x,y] - a[n]\, q_{1+m,1+n}[x,y] - a[m]\, q_{m,n}[x,y]\, q_{m,1+n}[x,y]\, q_{1+m,1+n}[x,y] - a[n]\, q_{m,n}[x,y]\, q_{m,1+n}[x,y]\, q_{1+m,1+n}[x,y] + a[m]\, q_{m,n}[x,y]\, q_{1+m,n}[x,y]\, q_{1+m,1+n}[x,y] + a[n]\, q_{m,n}[x,y]\, q_{1+m,n}[x,y]\, q_{1+m,1+n}[x,y] + a[m]\, q_{m,1+n}[x,y]\, q_{1+m,n}[x,y]\, q_{1+m,1+n}[x,y] - a[n]\, q_{m,1+n}[x,y]\, q_{1+m,n}[x,y]\, q_{1+m,1+n}[x,y]$$

Out[59]=

$$0$$

Out[60]=

$$0$$

In[61]:=
```
Collect[-F2, {q_{m,n}, q_{m+1,n+1}}, Factor]
```

Out[61]=

$$(a[m] + a[n]) \left(q_{m,1+n} - q_{1+m,n}\right) - (a[m] - a[n]) \left(1 + q_{m,1+n}\, q_{1+m,n}\right) q_{1+m,1+n} + q_{m,n} \left((a[m] - a[n]) \left(1 + q_{m,1+n}\, q_{1+m,n}\right) + (a[m] + a[n]) \left(q_{m,1+n} - q_{1+m,n}\right) q_{1+m,1+n}\right)$$